

ФЕДЕРАЛЬНОЕ АГЕНТСТВО
ОБРАЗОВАНИЯ РОССИЙСКОЙ ФЕДЕРАЦИИ
ДИМИТРОВГРАДСКИЙ ИНСТИТУТ ТЕХНОЛОГИЙ,
УПРАВЛЕНИЯ И ДИЗАЙНА

Лабораторная работа № 4

по Ассемблеру на тему:

"Программирование на уровне портов ввода-вывода"

Вариант № 16

*Выполнил студент группы ВТ-21: Потеренко А.Г.
Проверил преподаватель: Коноплянов А.В.*

Порядок работы.

1. Анализ индивидуального задания и разработка способов представления объектов задачи в памяти, методов доступа к ним.
2. Разработка программы на языке ассемблер.
3. Разработка контрольных примеров.
4. Отладка программ.
5. Составление отчета.

Содержание отчета.

	Стр.
1. Текст постановки задачи.....	3
2. Изложение способов представления объектов задачи в памяти и методов доступа к ним.....	3
3. Алгоритм.....	4
4. Описание и обоснование контрольных примеров.....	5
5. Текст программы с комментариями (в виде приложения).....	5-7

Отчет о выполненной работе.

1.1. Текст постановки задачи.

Необходимо написать программу, которая проигрывает мелодию через спикер в течение 20 секунд.

1.2. Общие требования задачи.

- Данная программа должна находиться в памяти, и перехватывать соответствующие прерывания (резидент).
- Возможность проверки копии резидента в памяти.
- Выгрузка резидента из памяти.

2. Изложение способов представления объектов задачи в памяти и методов доступа к ним.

Данная программа (резидент) представляет собой именно com-программу, так как прерывание INT 27h не может оставлять резидентными программы размером больше 64 Кбайт. Программа является активным резидентом, так как она перехватывает прерывание 9h от внешнего устройства (клавиатура). Данная com программа находится в памяти со смещением 100h относительно начала сегмента. После инсталляции резидента в памяти, программа должна освободиться от некоторой занимаемой памяти. Эта память включает в себя "Окружение DOS" и находится в PSP(префикс программного сегмента) по смещению 2ch. Также необходимо удалить часть программы за резидентом, то есть инсталлирующую часть программы, так как в основной работе резидента эта часть общей программы не требуется. В данной работе PSP сохраняется, за исключением окружения DOS-а.

Канал 2 порта 42h системного таймера управляет динамиком компьютера – он генерирует прямоугольные импульсы с частотой, равной 1193180/делитель_частоты Герц. После программирования канала 2 таймера надо еще включить сам динамик. Это осуществляется путем установки битов 0 и 1 порта 61h в 1. Бит 0 разрешает работу таймера, а бит 1 включает динамик. Также мы программируем порт 43h – управляющий регистр первого таймера. Порт 61h – включение и выключение динамика.

Через 60h-порт мы получаем байт данных от клавиатуры, то есть мы получаем скан-код нажатой клавиши, тем самым мы можем контролировать работу клавиатуры.

3. Алгоритм поставленной задачи.

АЛГОРИТМ УСТАНОВКИ РЕЗИДЕНТА В ПАМЯТИ

НАЧАЛО

ПРОВЕРКА НА НАЛИЧИЕ РЕЗИДЕНТА В ПАМЯТИ.

ЕСЛИ РЕЗИДЕНТ НАХОДИТСЯ В ПАМЯТИ ТО

ВЫДАЕМ СООТВЕТСТВУЮЩЕЕ СООБЩЕНИЕ

ИНАЧЕ

НАЧАЛО

УСТАНОВКА ВЕКТОРОВ ПРЕРЫВАНИЯ

УСТАНОВЛИВАЕМ РЕЗИДЕНТ

ДАЕМ ЕМУ УНИКАЛЬНЫЙ НОМЕР ОССН

УДАЛИТЬ ПЕРЕМЕННЫЕ ОКРУЖЕНИЯ DOS

УДАЛИТЬ ИНИЦИАЛИЗИРУЮЩУЮ ЧАСТЬ ИЗ ПАМЯТИ

ОСТАВИТЬ ПРОГРАММУ В ПАМЯТИ

КОНЕЦ

КОНЕЦ

АЛГОРИТМ ОБРАБОТКИ ПРЕРЫВАНИЯ INT2F-HANDLER

НАЧАЛО

ПРОВЕРИТЬ НАШ ЛИ ЭТО РЕЗИДЕНТ, ЕСЛИ НАШ, ТО ОН ДОЛЖЕН ИМЕТЬ НОМЕР ОССН

ЕСЛИ ДЕЙСТВИТЕЛЬНО НАШ РЕЗИДЕНТ ВЫЗВАЛ ЭТО ПРЕРЫВАНИЕ ТО AL=0FFH КОНЕЦ

ИНАЧЕ ПЕРЕХОД НА СТАРЫЙ ОБРАБОТЧИК 2FH

КОНЕЦ

АЛГОРИТМ ОБРАБОТКИ ПРЕРЫВАНИЯ INT9-HANDLER

НАЧАЛО

ПОЛУЧИТЬ БАЙТ ОТ ПОРТА КЛАВИАТУРЫ

ЕСЛИ НАЖАТА КЛАВИША F2 ТО

НАЧАЛО

УДАЛЕНИЕ ОБРАБОТЧИКА ИЗ ПАМЯТИ

ВОССТАНОВЛЕНИЕ ВЕКТОРОВ ПРЕРЫВАНИЯ

ПЕРЕХОД К СТАРОМУ ОБРАБОТЧИКУ ПРЕРЫВАНИЯ КЛАВИАТУРЫ

КОНЕЦ

ИНАЧЕ

ЕСЛИ НАЖАТА КЛАВИША F1 ТО ОБРАБОТКА ДИНАМИКА

ИНАЧЕ ПЕРЕХОД К СТАРОМУ ОБРАБОТЧИКУ ПРЕРЫВАНИЯ КЛАВИАТУРЫ

КОНЕЦ

4. Тестирование программы.

Программа запрограммирована таким образом, что при нажатии F1 звучит музыка через динамик, если резидент установлен и резидент выгружается из памяти, если нажата F2. Итак, рассмотрим различные варианты:

- Пользователь ввел в командной строке 4l.com. Пользователю выдается сообщение об успешной установке резидента в памяти. Для того чтобы убедиться, что резидент установлен, введем MEM /C в командной строке и просмотрим память.
Пользователь нажал на клавишу F1: нам прозвучала мелодия в течение 20 секунд.
Пользователь нажал на клавишу F2: резидент выгрузился из памяти. Проверка заключается в просмотре памяти (MEM /C – командная строка).
- Пользователь набрал в командной строке 4l.com, хотя резидент уже установлен в памяти. Пользователю выдается сообщение о том, что повторная установка копии резидента невозможна.

5. Текст программы с комментариями (в виде приложения).

```
.MODEL TINY
.CODE
ORG 2CH
DOS_OKRUGENIE DW ?
ORG 100H
;-----
;
; ТЕЛО ПРОГРАММЫ END
;-----
MAIN:
    JMP NEAR:INITIALIZE ;ПЕРЕХОД НА ИНИЦИАЛИЗИРУЮЩУЮ ЧАСТЬ
;-----
    TIME EQU 100 ;НЕКОТОРАЯ КОНСТАНТА РЕГУЛИРОВКИ ВРЕМЕНИ ЗВУЧАНИЯ МЕЛОДИИ
    REZ_ES DB 'РЕЗИДЕНТ "CSh" УЖЕ В ПАМЯТИ - НЕВОЗМОЖНО УСТАНОВИТЬ ЕГО КОПИЮ$'
    MEMORY DB 'ВВЕДИТЕ В КОМАНДНОЙ СТРОКЕ MEM /C:ОТОБРАЖЕНИЕ 4L',0DH,0AH,'$'
    REZIDENT_INSTALL DB 'РЕЗИДЕНТ УСПЕШНО УСТАНОВЛЕН',0DH,0AH,'$'
;-----ПАМЯТЬ ДЛЯ ХРАНЕНИЯ АДРЕСОВ-----
    OLD_SAVE_2FH DD 0 ;ПЕРЕМЕННАЯ ДЛЯ ХРАНЕНИЯ ОБРАБОТЧИКА
    OLD_SAVE_09H DD 0 ;ПЕРЕМЕННАЯ ДЛЯ ХРАНЕНИЯ ОБРАБОТЧИКА
;-----
;-----ОБРАБОТЧИК ПРЕРЫВАНИЯ ОТ КЛАВИАТУРЫ-----
;-----
INT9_HANDLER PROC FAR
;-----ПОЛУЧАЕМ БАЙТ ОТ ПОРТА КЛАВИАТУРЫ-----
    IN AL,60H
    CMP AL,3CH ;КНОПКА F2
    JNE NET@
;-----УДАЛЕНИЕ ОБРАБОТЧИКА ИЗ ПАМЯТИ-----
    XOR DX,DX
    MOV DS,DX ;ТАБЛИЦА ВЕКТОРОВ ПО ПАМЯТИ 0000h:0000h->DS=0
;--ДАЛЬШЕ ИДЕТ КРИТИЧЕСКИЙ УЧАСТОК-ЗАПОЛНЕНИЕ ТАБЛИЦЫ ВЕКТОРОВ--
    CLI ;ЗАПРЕТИТЬ ПРЕРЫВАНИЯ
    MOV AX,WORD PTR CS:[OLD_SAVE_09H]
    MOV DS:[09h*4],AX
    MOV AX,WORD PTR CS:[OLD_SAVE_09H+2]
    MOV DS:[09h*4+2],AX
;-----ВОССТАНАВЛИВАЕМ ВЕКТОР 2FH:ВДРУГ ЕСТЬ В ПАМЯТИ ЕЩЕ-----
;-----РЕЗИДЕНТЫ, ПЕРЕХВАТЫВАЮЩИЕ ТО ЖЕ ПРЕРЫВАНИЕ-----
    MOV AX,WORD PTR CS:[OLD_SAVE_2FH]
    MOV DS:[2Fh*4],AX
    MOV AX,WORD PTR CS:[OLD_SAVE_2FH+2]
    MOV DS:[2Fh*4+2],AX
    STI ;РАЗРЕШИТЬ ПРЕРЫВАНИЯ
;-----ОСВОБОЖДАЕМ ПАМЯТЬ - УДАЛЯЕМ БЛОК НАШЕГО КОДА-----
    PUSH CS
    POP ES ;ES-СЕКМЕНТНЫЙ АДРЕС ОСВОБОЖДАЕМОГО БЛОКА
    MOV AX,4900h
    INT 21h ;ОСВОБОДИТЬ ПАМЯТЬ
;-----ПЕРЕХОДИМ К СТАРОМУ ОБРАБОТЧИКУ ПРЕРЫВАНИЯ-----
    JMP DWORD PTR CS:OLD_SAVE_09H
```

```

;-----ОБРАБОТКА ДИНАМИКА-----
NET@:
;-----ПОЛУЧАЕМ БАЙТ ОТ ПОРТА КЛАВИАТУРЫ-----
IN AL,60H
CMP AL,3BH
JNE NET ;ЕСЛИ ЭТО НЕ F1
;-----РАБОТА С ДИНАМИКОМ НА УРОВНЕ ПОРТОВ-----
;7,6 БИТЫ=10:КАНАЛ 2
;5,4 БИТЫ=11:ЧТЕНИЕ МЛАДШЕГО, ПОТОМ СТАРШЕГО БАЙТА
;3-1 БИТЫ=011:ГЕНЕРАТОР ПРЯМОУГОЛЬНЫХ ИМПУЛЬСОВ (ОСНОВНОЙ РЕЖИМ)
;0 БИТ=0: ФОРМАТ СЧЕТЧИКА - ДВОИЧНОЕ 16-БИТНОЕ ЧИСЛО
MOV AL,10110110B
;ЗАПИШЕМ ДАННЫЕ В ПОРТ
OUT 43H,AL ;ПОРТ 43H-УПРАВЛЯЮЩИЙ РЕГИСТР 1-ГО ТАЙМЕРА
;-----ВКЛЮЧЕНИЕ ДИНАМИКА-----
;ЗАПИШЕМ ТЕКУЩЕЕ СОСТОЯНИЕ ПОРТА В AL
IN AL,61H
OR AL,00000011B ;УСТАНОВИМ БИТЫ 0 И 1 В 1-ЦУ
OUT 61H,AL ;ВКЛЮЧИМ ДИНАМИК-ЗАНЕСЕМ ДАННЫЕ В ПОРТ
;-----ВКЛЮЧИТЬ ДИНАМИК И ПРОПЕТЬ МЕЛОДИЮ-----
MOV CX,TIME
MOV BL,90 ;МЛАДШИЙ БАЙТ ДЕЛИТЕЛЯ ЧАСТОТЫ
MOV BH,6 ;СТАРШИЙ БАЙТ ДЕЛИТЕЛЯ ЧАСТОТЫ
SIKL_DINAMIC:
;-----
MOV AL,BL
OUT 42H,AL ;ЗАНОСИМ МЛАДШИЙ БАЙТ ДЕЛИТЕЛЯ ЧАСТОТЫ
MOV AL,BH
OUT 42H,AL ;ЗАНОСИМ СТАРШИЙ БАЙТ ДЕЛИТЕЛЯ ЧАСТОТЫ
ADD BL,5
;-----
PUSH CX
MOV CX,64000
ZADERGKA_TIME:
PUSH CX
MOV CX,0FFF0H
MOV DX,0AFFFH
MOV Ah,86H
INT 15H ;ЗАДЕРЖАЛИ ВРЕМЯ
POP CX
LOOP ZADERGKA_TIME
POP CX
;-----
LOOP SIKL_DINAMIC
;-----НАСТАЛО ВРЕМЯ ВЫКЛЮЧИТЬ ДИНАМИК-----
IN AL,61H
AND AL,11111100B ;ОБНУЛЯЕМ МЛАДШИЕ 2 БИТА
OUT 61H,AL ;ЗАНОСИМ ДАННЫЕ В ПОРТ
;-----
;-----ПЕРЕХОД НА СТАРЫЙ ОБРАБОТЧИК-----
NET:
JMP DWORD PTR CS:OLD_SAVE_09H
INT9_HANDLER ENDP
;-----
;-----МУЛЬТИПЛЕКСОРНОЕ ПРЕРЫВАНИЕ-ПРЕРЫВАНИЕ ПОЛЬЗОВАТЕЛЯ-----
;-----ОБРАБОТКА ПРЕРЫВАНИЯ INT 2FH-----
INT2F_HANDLER PROC FAR
CMP AH,0CCH
JNE END_2FH
CMP AL,0
JNE END_2FH
;---ДА, ДЕЙСТВИТЕЛЬНО ОБРАЩЕНИЕ ИМЕННО К НАШЕМУ ОБРАБОТЧИКУ-----
MOV AL,0FFH ;ЗАПОМИНАЕМ - ИМЕННО НАШ РЕЗИДЕНТ НАХОДИТСЯ В ПАМЯТИ
IRET ;ВОЗВРАТ ИЗ ОБРАБОТЧИКА ПРЕРЫВАНИЯ
INT2F_HANDLER ENDP
;-----
END_2FH:

```

```

JMP CS:OLD_SAVE_2FH      ;ОТПРАВЛЯЕМСЯ НА ОБРАБОТКУ СТАРОГО ПРЕРЫВАНИЯ
;-----
;
;
;
;
;
;
;-----
PROCEDURA, УСТАНОВЛИВАЮЩАЯ ОБРАБОТЧИК
;-----
INITIALIZE PROC NEAR
;-----ПРОВЕРКА, ЕСЛИ РЕЗИДЕНТ В ПАМЯТИ УЖЕ-----
;/////////////////////////ВХОДНЫЕ ДАННЫЕ/////////////////////////
MOV AH,0CCH ;РЕЗИДЕНТ ПОЛУЧАЕТ УНИКАЛЬНЫЙ НОМЕР
MOV AL,0      ;ПРОВЕРКА НАЛИЧИЯ РЕЗИДЕНТА В ПАМЯТИ
INT 2FH ;ПРОВЕРЯЕМ НА ПОВТОРНУЮ УСТАНОВКУ
;/////////////////////////ВЫХОДНЫЕ ДАННЫЕ/////////////////////////
CMP AL,0FFH ;ПРОВЕРКА,ЕСЛИ В AL-Е ЛЕЖИТ 2FFH,ТО РЕЗИДЕНТ УЖЕ УСТ.
JNE INSTALL
;-----ИНАЧЕ-ВЫХОД ИЗ ПРОГРАММЫ-----
MOV AX,0900H ;ВЫВОД СООТВЕТСТВУЮЩИХ СООБЩЕНИЙ
LEA DX,MEMORY
INT 21H
LEA DX,REZ_ES
INT 21H
RET ;ВЫХОД ИЗ 4L.COM
INSTALL:
;-----ЕСЛИ НЕТ, ТО УСТАНОВЛИВАЕМ РЕЗИДЕНТ-----
MOV AX,0900H ;ВЫВОД СООТВЕТСТВУЮЩИХ СООБЩЕНИЙ
LEA DX,REZIDENT_INSTALL
INT 21H
;-----ИСПОЛЬЗУЕМ МУЛЬТИПЛЕКСОРНОЕ ПРЕРЫВАНИЕ-----
;-----ПОЛУЧАЕМ И СОХРАНЯЕМ АДРЕС СТАРОГО ОБРАБОТЧИКА-----
MOV AX,352FH
INT 21H
MOV WORD PTR OLD_SAVE_2FH,BX
MOV WORD PTR OLD_SAVE_2FH+2,ES
;-----
;-----УСТАНОВИТЬ НОВЫЙ ОБРАБОТЧИК ПРЕРЫВАНИЯ 2FH-----
;-----
MOV AX,252FH
LEA DX,INT2F_HANDLER
INT 21H
;-----
;-----КЛАВИАТУРА-----
;-----
;-----ПОЛУЧАЕМ И СОХРАНЯЕМ АДРЕС СТАРОГО ОБРАБОТЧИКА-----
MOV AX,3509H
INT 21H
MOV WORD PTR OLD_SAVE_09H,BX
MOV WORD PTR OLD_SAVE_09H+2,ES
;-----УСТАНОВИТЬ НОВЫЙ ОБРАБОТЧИК ПРЕРЫВАНИЯ КЛАВИАТУРЫ-----
MOV AX,2509H
LEA DX,INT9_HANDLER
INT 21H
;-----
;-----УДАЛИТЬ ПЕРЕМЕННЫЕ ОКРУЖЕНИЯ DOS-----
;-----
MOV AX,4900H
MOV ES,WORD PTR DOS_OKRUGENIE
INT 21H
;-----УДАЛИТЬ ЧАСТЬ ПРОГРАММЫ ПОСЛЕ РЕЗИДЕНТА-----
LEA DX,INITIALIZE
INT 27H
INITIALIZE ENDP
;-----
END MAIN

```